Memoirs of the School of Engineering, Okayama University Vol.13, February 1979

Computer Program of Line Balancing, Regarding Efficiency and Number of Stations as Variables

Fumio AKAGI*, Hirokazu OSAKI*, and Susumu KIKUCHI*

(Received February 6, 1979)

Synopsis

To assign work elements to the work stations in an assembly or manufacturing line, various computer programs have been developed and used. And it does that the number of stations or the cycle time is even given. But in practice it is desirable to obtain the assignment which shows the highest efficiency of line balancing under all possible combinations of the number of stations and the cycle time.

Therefore we propose a computer program of the assignment method in which the efficiency of line balancing, Ebb and the number of stations, NN are regarded as variables. In this method the minimum value (EEb) of efficiency and the constant term (d) by which Ebb is reduced are given previously. And for any combination of Ebb ($EEb \leqq Ebb \leqq 100$) and NN ($1 \leqq NN \leqq Nm$: Nm calculated from EEb), the work elements are assigned to work stations, the precedence restrictions being used. Ebb is reduced by d from the ideal value (100) until the assignment to NN is obtained. The efficiency of the obtained assignment, Ebo is calculated.
As $Ebb < Ebo \leqq Ebb+d$, the calculation is continued until the assignment to NN, which shows the maximum efficiency, is obtained. In this process NN varies from Nm to 1 by 1 by 1.

* Department of Industrial Science, Scool of Engineering Okayama University.

1. Introduction

The basic problem in line balancing is to assign the work elements to the work stations so as to balance the workload (the station time) at each station and to make this work content as close to the cycle time of the line as possible [1]. Various computer programs have been developed and used to solve this problem [2,3]. Generally to evaluate the assignment, the efficiency of line balancing, which is indicated by the following Eb, has been used [4].

$$Eb=\frac{T}{N \times P} \times 100 \qquad (1)$$

where Eb = the efficiency of line balancing

T = the total work time

N = the number of stations

P = the cycle time to N stations.

As there is no slack time in the line at Eb=100%, the assignment of the work elements is ideal.

In the programs developed until now, N or P must be given. As N is given, the assignment is determined so as to minimize P. As P is given, it is done so as to minimize N. The number of stations N is determined by personel, machines, tools, work areas, and so on. The cycle time P is restricted by the production time and the units to be produced per day. From these view points the number of stations (NN), the efficiency of line balancing (Ebb), and the cycle time (PP) are regarded as variables in some restrictions. Further there is a functional relation among NN, PP and Ebb.

Therefore in this paper we propose an assignment method in which the work elements are assigned to the work stations, Ebb and NN regarded as variables. Firstly, in this method the minimum efficiency of line balancing (EEb) and the constant term (d) by which Ebb is reduced are given. Then the maximum number of stations (Nm) is determined by EEb and the maximum work element time (tm) from Eq.(1), and the value of NN is given as Nm. Next, the value of Ebb is given as 100 (the ideal value). Then PP is determined from Ebb and NN from Eq. (1). NN and PP being used, the work elements are assigned to the work stations according to the precedence restrictions so that the work time of each station (the station time) may be less than PP. Ebb is reduced at the constant rate (d) from 100 until the assignment to NN is obtained. And the efficiency of the obtained assignment (Ebo) is calculated. Ebo is greater than Ebb but less than Ebb+d because the maximum station time (C) is less than PP. Since then Eb is made exchange Ebb to Ebo.

And the calculation is continued until the assignment to NN which shows the maximum of Eb between Ebb and Ebb+d is obtained. The same procedures are repeated from Nm to 1 by 1 by 1.

As this method needs much repetition of the routine calculation, we develop the computer program.

2. Procedures for Assigning Work Elements to Work Stations

In order to obtain a favorable assignment, the sequence of the operations must be analyzed first. The precedence restrictions are used to indicate which work element must be done before others. The results are summarized by a precedence diagram[5], an arrow diagram[6], a table of functional units[7], a precedence matrix[1], and so on.

The proposed procedure for assigning the work elements to the work stations using the precedence restrictions is as follows.

Step 1. EEb and d are given.

Step 2. The theoretical maximum number of stations, Nm is calculated from

$$Nm = \left[\frac{T}{tm} \times \frac{100}{EEb}\right] \qquad (2)$$

where ti = the operation time of work element i

tm = max ti, the maximum work element time

T = $\sum$ti, the total work time

[ ] : Gaussian symbol.

Therefore the possible range of NN is from Nm to 1, that is, $1 \leqq NN \leqq Nm$. For any NN, the following procedures are applied.

Step 3. At the first time NN is given as Nm. After the second time NN is down from Nm by 1, that is, NN=NN-1. Go to next step.

When NN equals 1, stop the procedure.

Step 4. PP to NN and Ebb is calculated by

$$PP = \frac{T}{NN} \times \frac{100}{Ebb} . \qquad (3)$$

Ebb is reduced by d from 100 until an assignment to NN is obtained. To any combination (NN,PP), go to next step.

Step 5. At the first time, select the work elements which don't have the precedence elements.

After the second time, add to them the work elements which have only the assigned ones as the precedence ones. Go to next step.

Step 6. Select the work elements, which is less than the remaining cycle time (the slack time) of the station, from among the work elements selected at step 5. Then go to step 7.

If no work element can be selected here, proceed to the next station and repeat the same step.

If the last station has been examined, go to step 8.

Step 7. Assign the work element having the maximum work time in those elements selected at step 6 to the station. Subtract the work time from the remaining cycle time and eliminate the assigned work element. Then return step 5.

Step 8. When all the elements has been assigned to all the stations, an assignment has been obtained.

The efficiency of the obtained assignment (Ebo) is calculated from

$$Ebo = \frac{T}{NN \times C} \times 100 \qquad (4)$$

where C = the maximum station time of the obtained assignment for NN and PP

$C < PP$.

Because of $Ebb < Ebo \leq Ebb + d$, since then we search the assignment which shows the maximum efficiency. That is, let Ebb = Ebo and PP = C. To the new combination of (NN,PP), return step 5.

If one or more work elements remain and an assignment to NN hadn't been obtained, return step 4 and reduce Ebb by d, that is, Ebb = Ebb-d.

If one or more work elements remain and an assignment to NN had been ever obtained, return step 3 and reduce NN by 1.

3. Program

This program is written in Fortran IV and is the form of the subroutine.

The subroutine name is CPLB.

```
SUBROUTINE CPLB(EEB,D,NWORK,NAME,TIME,NA0,KINDP,NAMEP,NSTART)
```

3.1. Usage

The work time and the precedence restrictions among work elements are provided from the table of functional units or the arrow diagram.

3.1.1 Argument List in the case of Table of Functional Units

ARGUMENT	I/O	TYPE	SIZE	DEFINITION
EEB	I	REAL	1	EEb, the minimum efficiency of line balancing
D	I	REAL	1	d, the constant term by which Ebb is reduced
NWORK	I	INTEGER	1	number of work elements
NAME	I	CHARACTER	50X5	name of work element (5A4)

ARGUMENT	I/O	TYPE	SIZE	DEFINITION
TIME	I	REAL	50	ti, time of work element
NA0	I	INTEGER	50	work element's number
KINDP	I	INTEGER	50	number of parts in work element
NAMEP	I	INTEGER	50	part's number in work element
NSTART	I	INTEGER	1	0

where C < D < EEB ≦ 100, NWORK ≦ 50, NAME ≦ 20 characters, KINDP ≦ 10

3.1.2 Argument List in the case of Arrow Diagram

ARGUMENT	I/O	TYPE	SIZE	DEFINITION
EEB	I	REAL	1	EEb, the minimum efficiency of line balancing
D	I	REAL	1	d, the constant term by which Ebb is reduced
NWORK	I	INTEGER	1	number of work elements
NAME	I	CHARACTER	50X5	name of work element (5A4)
TIME	I	REAL	50	ti, time of work element
NA0	I	INTEGER	50	preceding event number
KINDP	I	INTEGER	50	succeeding event number
NAMEP	I	INTEGER	50	not use in this case
NSTART	I	INTEGER	1	source node number

where 0 < D < EEB ≦ 100, NWORK ≦ 50, NAME ≦ 20 characters.

3.2. Suggestion on Using

Subroutine GRPUNT or GRPARR, PRINT, and MAXGRP are used in CPLB. GRPUNT is used to select work elements which don't have the precedence works using the table of functional units. GRPARR to do so using the arrow diagram. PRINT to print out the results of the assignment. And MAXGRP to select the work element having the maximum work time from the selected work elements.

Program list is shown in Table 1.

4. Example

The assembly work of a small electric switch is used as an example to excute the program. The work has been analyzed and divided into work elements. The table of the functional units and the list of the arrow diagram have been developed.

Either the table or the list is used as the input data of the precedence restrictions among the work elements. The computer outputs are the same.

The data given from the table of the functional units are shown in Table 2. And the data given from the list of the arrow diagram are shown in Table 3. The computer outputs of this example are shown in Table 4.

References

[1] ASTME : "Manufacturing Plannıng and Estimating Handbook", McGraw-Hill(1963), 9-41.
[2] K. Takeda : IE, 5(1975), 39.
[3] E. M. Darel : AIIE Trans., 7-3(1975), 302.
[4] M. Kawashima : "Shin Sagyo to Layout", Nihon Noritsu Kyokai(1964) 199.
[5] T. O. Prenting and R. M. Battglin : J. Indust. Engng., 15-4(1964-7), 208.
[6] K. Tone : "PERT Nyumon", Toyo keizai(1973), 18.
[7] F. Akagi, H. Osaki and S. Kikuchi : Conference Report of J. S. M. E., 780-8(1978-4), 24.

Table 1. Program Listing

```
      SUBROUTINE CPLB(EEB,D,NWORK,NAME,TIME,NAO,KINDP,NAMEP,NSTART)
      DIMENSION NAME(50,5),TIME(50),NAO(50),KINDP(51),NAMEP(50,10)
      DIMENSION NGROUP(50),NSGRP(50),NSGRP1(50),NAOS(50),NAOSS(50)
      DIMENSION CYCLET(50),NKELMT(50),NSTELM(50,50)
      BIG=0.0
      SUM=0.0
      DO 100 I=1,NWORK
      SUM=SUM+TIME(I)
      IF(TIME(I) .LE. BIG) GO TO 100
      BIG=TIME(I)
  100 CONTINUE
      STN=SUM/EEB/BIG*100.0
      NSTN=STN
   90 CONTINUE
      IF(NSTN .LE. 0) GO TO 91
      EB=1.0
      KDOWN=0
      TMEAN=SUM/FLOAT(NSTN)
   55 CONTINUE
      TMAX=TMEAN/EB
   50 CONTINUE
      TMIN=TMEAN*(2.0-1.0/EB)
      DO 101 I=1,NWORK
       NAOS(I)=NAO(I)
       NAOSS(I)=NAO(I)
  101 CONTINUE
      DO 110 I=1,NSTN
       CYCLET(I)=0.0
       NKELMT(I)=0
  110 CONTINUE
      K=0
      NUMGRP=0
      NPLACE=0
      MAXET=51
      KINDP(51)=NSTART
   10 CONTINUE
      K=K+1
      IF(K .GT. NSTN) GO TO 30
      L=0
   20 CONTINUE
      IF(NSTART .LE. 0) GO TO 1
      CALL GRPARR(NWORK,NUMGRP,NGROUP,NAOS,NAOSS,KINDP,NAO,MAXET)
      GO TO 2
    1 CONTINUE
      CALL GRPUNT(NWORK,NUMGRP,NGROUP,NAOS,NAOSS,KINDP,NAMEP)
    2 CONTINUE
      DO 250 I=1,NUMGRP
      NUM=NGROUP(I)
      IF(NAOS(NUM) .LE. 0) GO TO 250
      IF(TIME(NUM) .LE. 0.0) GO TO 61
  250 CONTINUE
      K1=0
```

```
      TREST=TMAX-CYCLET(K)
      NTREST=TREST*10000.0
      TREST=NTREST/10000
      DO 300 I=1,NUMGRP
       K2=NGROUP(I)
       IF(NAOS(K2) .EQ. 0) GO TO 300
       IF(TREST .LE. TIME(K2)) GO TO 300
       K1=K1+1
       NSGRP(K1)=NGROUP(I)
  300 CONTINUE
      IF(K1 .EQ. 0) GO TO 10
      MAXET=MAXGRP(K1,NSGRP,TIME)
      CYCLET(K)=CYCLET(K)+TIME(MAXET)
      NKELMT(K)=NKELMT(K)+1
      L=NKELMT(K)
      NSTELM(K,L)=MAXET
      GO TO 60
   61 MAXET=NUM
   60 CONTINUE
      NAOS(MAXET)=0
      NPLACE=NPLACE+1
      GO TO 20
   30 CONTINUE
      IF(NPLACE .GE. NWORK) GO TO 41
      IF(KDOWN .NE. 0) GO TO 40
      EB=EB-(D/100.0)
      GO TO 55
   41 CONTINUE
      CALL PRINT(SUM,NSTN,CYCLET,NKELMT,NSTELM,NAO,NAME,TIME,EB,TMAX,
     &           TMIN,TMEAN)
      KDOWN=1
        GO TO 50
   40 CONTINUE
      WRITE(6,6800)
 6800 FORMAT(////,20X,'DIVIDE THE WORK ELEMENT OR REDUCE THE STATION',)
      NSTN=NSTN-1
      GO TO 90
   91 CONTINUE
      RETURN
      END


C **  SELECT ASSIGNABLE WORK ELEMENT FROM TABLE OF FUNCTIONAL UNITS   **
      SUBROUTINE GRPUNT(NWORK,NUMGRP,NGROUP,NAOS,NAOSS,KINDP,NAMEP)
      DIMENSION NGROUP(50),NAOS(50),NAOSS(50),KINDP(51),NAMEP(50,10)
      DO 200 I=1,NWORK
        IF(NAOSS(I) .EQ. 0) GO TO 200
        KP=KINDP(I)
        DO 210 J=1,KP
         DO 211 I1=1,NWORK
          IF(NAOS(I1) .EQ. 0) GO TO 211
          IF(NAMEP(I,J) .EQ. NAOS(I1)) GO TO 200
  211    CONTINUE
  210   CONTINUE
        NUMGRP=NUMGRP+1
        NGROUP(NUMGRP)=I
        NAOSS(I)=0
  200 CONTINUE
      RETURN
      END
```

```
C **  SELECT ASSIGNABLE WORK ELEMENT FROM ARROW DIAGRAM   **
      SUBROUTINE GRPARR(NWORK,NUMGRP,NGROUP,NAOS,NAOSS,NBACK,NFRONT,
     &                       MAXET)
      DIMENSION NGROUP(50),NAOS(50),NAOSS(50),NBACK(51),NFRONT(50)
      DO  100 I=1,NWORK
      IF(NAOS(I) .LE. 0) GO TO 100
      IF(NBACK(I) .EQ. NBACK(MAXET)) GO TO 300
  100 CONTINUE
      DO 200 I=1,NWORK
      IF(NAOSS(I) .LE. 0) GO TO 200
      IF(NFRONT(I) .NE. NBACK(MAXET)) GO TO 200
      NUMGRP=NUMGRP+1
      NGROUP(NUMGRP)=I
      NAOSS(I)=0
  200 CONTINUE
  300 CONTINUE
      RETURN
      END

C **  PRINT OUT THE SOLUTION OF ASSIGNMENT   **
      SUBROUTINE PRINT(SUM,NSTN,CYCLET,NKELMT,NSTELM,
     &             NAO,NAME,TIME,EB,TMAX,TMIN,TMEAN)
      DIMENSION CYCLET(50),NKELMT(50),NSTELM(50,50),NAO(50),
     &            NAME(50,10),TIME(50)
      WRITE(6,6100) NSTN,EB,SUM,TMEAN,TMAX,TMIN
 6100 FORMAT(1H ,////,30X,'THE INITIAL DATA',/,10X,'NUM. OF STATION =',
     1        I5,5X,'EB=',F5.3,5X,'SUM=',F10.5,2X,'TMEAN=',F10.5,2X,
     2        'TMAX=',F10.5,2X,'TMIN=',F10.5,//,10X,'STATION',3X,
     3        'CYCLE TIME',3X,'WORK ELEMENT',/)
      BIG=0.0
      DO 600 I=1,NSTN
      WRITE(6,6200) I,CYCLET(I),NKELMT(I)
 6200 FORMAT(10X,I5,3X,F10.5,I5)
      L=NKELMT(I)
      DO 610 J=1,L
      J2=NSTELM(I,J)
      WRITE(6,6300) NAO(J2),(NAME(J2,J3),J3=1,4),TIME(J2)
 6300 FORMAT(30X,I5,5X,4A4,F10.4)
  610 CONTINUE
      IF(CYCLET(I) .LE. BIG) GO TO 600
      BIG=CYCLET(I)
  600 CONTINUE
      EB=SUM/(NSTN*BIG)
      AG=FLOAT(NSTN)*BIG/SUM-1.0
      WRITE(6,6400) BIG,EB,AG
 6400 FORMAT(//,20X,'TMAX=',F10.5,5X,'EB=',F5.3,5X,'AG=',F5.3)
      TMAX=BIG
      RETURN
      END

      FUNCTION MAXGRP(NUMBER,NSET,DATA)
      DIMENSION NSET(50),DATA(50)
      BIG=0.0
      DO 100 I=1,NUMBER
       NUM=NSET(I)
       IF(DATA(NUM) .LE. BIG) GO TO 100
       MAXGRP=NUM
       BIG=DATA(NUM)
  100 CONTINUE
      RETURN
      END
```

Table 2. Given Data from Table of Functional Units

EEB = 80.0

D = 5.0

NWORK= 6

NAME(i,j)	TIME(i)	NA0(i)	KINDP(i)	NAMEP(i,k)
BASE	32.3	1	7	101 802 803 804 805 306 407
GUIDE	15.3	2	4	802 408 309 810
BUTTON	5.8	3	3	811 2 803
TERMINAL 1	12.6	4	3	112 313 805
TERMINAL 2	12.6	5	3	112 313 805
TERMINAL 3	12.6	6	3	112 313 805

NSTART= 0

Table 3. Given Data from List of Arrow Diagram**

EEB = 80.0

D = 5.0

NWORK= 9

NAME(i,j)	TIME(i)	NA0(i)	KINDP(i)
BASE	32.3	1	6
GUIDE	15.3	1	2
BUTTON	5.8	2	6
TERMINAL 1	12.6	1	3
TERMINAL 2	12.6	1	4
TERMINAL 3	12.6	1	5
DUMMY	0.0	3	6
DUMMY	0.0	4	6
DUMMY	0.0	5	6

NSTART= 1

** Arrow Diagram of the Small Electric Switch



Table 4. Computer Output

```
                        THE INITIAL DATA
NUM. OF STATION =        3 *1  EB=0.900 *2   SUM= *3 91.20000   TMEAN=   30.4000

STATION    CYCLE TIME    WORK ELEMENT        TMAX= *4 33.77778  TMIN=    27.0222

   1       32.30000       1
                            1       BASE                  32.3000 *5
   2       27.90000       2
                            2       GUIDE                 15.3000
                            4       TERMINAL 1            12.6000
   3       31.00000       3
                            5       TERMINAL 2            12.6000
                            6       TERMINAL 3            12.6000
                            3       BUTTON                 5.8000

           TMAX=   32.30000 *6   EB=0.941 *7    AG=0.062  *8

                        THE INITIAL DATA
NUM. OF STATION =        2      EB=0.950     SUM=  91.20000   TMEAN=   45.6000

STATION    CYCLE TIME    WORK ELEMENT        TMAX=  48.00000  TMIN=    43.2000

   1       44.90000       2
                            1       BASE                  32.3000
                            4       TERMINAL 1            12.6000
   2       46.30000       4
                            2       GUIDE                 15.3000
                            5       TERMINAL 2            12.6000
                            6       TERMINAL 3            12.6000
                            3       BUTTON                 5.8000

           TMAX=   46.30000      EB=0.985       AG=0.015

                        THE INITIAL DATA
NUM. OF STATION =        1      EB=0.950     SUM=  91.20000   TMEAN=   91.2000
                                             TMAX=  96.00000  TMIN=    86.4000
STATION    CYCLE TIME    WORK ELEMENT

   1       91.20000       6
                            1       BASE                  32.3000
                            2       GUIDE                 15.3000
                            4       TERMINAL 1            12.6000
                            5       TERMINAL 2            12.6000
                            6       TERMINAL 3            12.6000
                            3       BUTTON                 5.8000

           TMAX=   91.20000      EB=1.000       AG=*****
```

*1 $N_m = NN$, *2 $E_{bb}/100$, *3 $T = \sum t_i$, *4 PP, *5 $t_m = \max t_i$,

*6 C ,

*7 $E_{bo}/100 = T/(NN \times C)$,

*8 $AG = (NN \times C - T)/T$.